

```

// T45mixDesRay.sce
// MIXTURE ESTIMATION (descriptive, Rayleigh)
// - static componens
// Experiments
// - change simulated parameters
// - change initial parameters
// -----
exec("ScIntro.sce",-1),
getd(), mode(0)

nd=500; // 1
// PARAMETERS // 2
rS=[10 30 20]; // simulated comp. expectations // 3
nc=length(rS); // number of components // 4
aS=[.3 .3 .4]; // parameters of pointer model // 5
// 6
// SIMULATION // 7
for t=1:nd // 8
    jS=sampCat(aS); // pointer value // 9
    cS(t)=jS; // stor pointer value // 10
    y(t)=sqrt(-2*rS(jS)*log(randu())); // output // 11
end // 12
// 13

```

```

// ESTIMATION // 14
// initialization // 15
rE=[10 15 30]; // initial component parameters // 16
ka=[1 1 1]; // initial counter // 17
S=2*ka.*rE; // statistics corresponding to // 18
// to the parameters // 19

// time loop of estimation // 20
th=rE; // 21
for t=1:nd // 22
    for j=1:nc // 23
        q(j)=y(t)/rE(j)*exp(-y(t)^2/(2*rE(j))); // proximity // 24
    end // 25
    w=q/sum(q); // weights // 26
    wt(:,t)=w; // remember weights // 27
    for j=1:nc // 28
        S(j)=S(j)+w(j)*y(t)^2; // update of inf. matrix // 29
        ka(j)=ka(j)+w(j); // update of counter // 30
        rE(j)=S(j)/(2*ka(j)); // point estimates // 31
    end // 32
    th=[th; rE]; // remember point estimates // 33
end // 34
// 35
// RESULTS // 36

```

```

tx=['b','r','g','k']; // 37
set(scf(1),'position',[600 10 600 400]) // evolution of par. est. // 38
title 'Evolution of the estimated parameters' // 39
for j=1:nc // 40
    plot(th(:,j),'-'+tx(j)) // 41
end // 42
legend('c1','c2','c3','c4'); // 43
// 44
disp 'The final parmeter estimates are' // 45
disp(rE) // 46
// 47
[nill,cp]=max(wt,'r'); // accuracy of classification // 48
disp 'Accuracy of classification' // 49
ACC=acc(cS,cp) // 50

```

## Description of the program

The distribution of the components (denoted by  $j$ ) is exponential one

$$f_j(y_t|r_j) = \frac{y_t}{r_j} \exp\left(-\frac{y_t^2}{2r_j}\right)$$

For the definition of statistics, their update and computation of parameter estimates see Section XXX.

- Rows 3–5 define model parameters for simulation.
- Rows 8–12 perform output simulation
- Rows 16–18 set initial parameters and corresponding statistics.
- Rows 21–35 perform the time loop for estimation.
  - Rows 23–25 compute component proximities.
  - Rows 26–27 construct the weights  $w$ .
  - Rows 28–32 perform statistics update and construct estimates of the parameters.