
function [x,Rx,yp] = Kalman(x,y,ut,M,N,F,A,B,G,Rw,Rv,Rx)

// [x,Rx,yp]=Kalman(x,y,ut,M,N,F,A,B,G,Rw,Rv,Rx)

// Kalman filter for state estimtion with the model

// x = M*x + N*ut + F + w

// yp = A*x + B*ut + G + e

// x state

// Rx state estimate covariance matrix

// yp output prediction

// y output

// ut input

// M,N,F state model parameters

// A,B,G output model parameters

// Rw state model covariance

// Rv output model covariance

nx=size(M,1); // 1

ny=size(A,1); // 2

if isempty(F) then F=zeros(nx,1);end, // 3

if isempty(G) then G=zeros(ny,1);end, // 4

// 5

x=M*x+N*ut+F; // time update of the state // 6

Rx=Rw+M*Rx*M'; // time updt. of state covariance // 7

// 8



yp=A*x+B*ut+G; // output prediction // 9

Ry=Rv+A*Rx*A'; // noise covariance update // 10

Rx=Rx-Rx*A'*inv(Ry)*A*Rx; // state est. covariance update // 11

ey=y-yp; // prediction error // 12

KG=Rx*A'*inv(Rv); // Kalman gain // 13

x=x+KG*ey; // data update of the state // 14

endfunction // 15

Description of the program

This procedure arises from the general scheme of state evolution when we substitute normal distributions for individual

pdfs. The normality then is preserved during the evolution and the relation for pdfs is converted to that for expectations

and covariances.

• Row 6 predicts new value of the state (without using data).

• Row 7 construct covariance of the new state estimate .

• Row 9 performs prediction of the output (based only on the model, not new data).

• Row 10 computes covariance of the performed output prediction.

• Row 11 recomputes covariance of the state taking into account measured data.

• Row 12 construct prediction error - this error speaks about the correspondence of the state estimate and the

reality in the data.



• Row 13 constructs so called Kalman gain - it is an ampli�cation factor through which the existing state estimate

is corrected through the prediction error.

• Row 14 gives the �nal corrected state estimate in the actual time instant.


