

```

// T11simReg4.sce
// MULTIVARIATE REGRESSION MODEL AND BAYESIAN ESTIMATION
// Experiments
// - change parameters of the model
// - change the input signal
// - try to increase the model order to 2
// -----
exec("ScIntro.sce",-1), mode(0)

// PARAMETERS // 1
nd=100; // length of data // 2
a1S=[.8 .2 // 3
     .1 .6]; // parameters at y(t-1) // 4
b0S=[1 -.5]'; // parameters at u(t) // 5
b1S=[.5 .1]'; // parameters at u(t-1) // 6
kS=[-5 3]'; // constant (model absolute term) // 7
sS=.1*eye(2,2); // noise std // 8
ny=max(size(a1S)); // number of y-variables // 9
// 10
y(1,:)= [1 -1]; // initial conditions for output // 11
u=signal(nd,1,.1); // input // 12
// 13
// TIME LOOP // 14

```

```

thS=[b0S a1S b1S kS]';           // matrix of parameters           // 15
V=zeros(7,7);                     // initial statistics           // 16
ka=0;                             // 17
for t=2:nd                         // 18
    ps=[u(t) y(t-1,:) u(t-1) 1]'; // regression vector           // 19
    y(t,:)=thS'*ps+sS*rand(2,1,'n'); // simulation                   // 20
    Ps=[y(t,:) ps']';             // extended regression vector   // 21
    V=V+Ps*Ps';                  // update of statistics         // 22
    ka=ka+1;                      // 23
end                                // 24
Vy=V(1:ny,1:ny);                 // partitioning                 // 25
Vyp=V(ny+1:$,1:ny);              // of information               // 26
Vp=V(ny+1:$,ny+1:$);             // matrix                       // 27
thE=inv(Vp)*Vyp;                 // parameter estimates         // 28
// RESULTS                        // 29
// RESULTS                        // 30
set(gcf(),'position',[700 100 600 500]) // 31
subplot(211),plot(1:nd,u),title('Input') // 32
subplot(212),plot(1:nd,y),title('Output') // 33
disp('The simulated and estimated parameters are',[thS,thE]) // 34

```

Description of the program

- Rows 3–8 set parameters of the model
- Row 11 sets the initial conditions
- Row 12 generates input signal
- Rows 16–18 are initial statistics (no prior information)
- Rows 18–24 perform the time loop
 - 19: construction of regression vector
 - 20: simulation
 - 21: construction of extended regression vector
 - 22–23: update of statistics
- Rows 25–27 divides the information matrix
- Row 28 computes point estimates of the parameters